
    nx_tflite_mob

Call TensorFlow Lite models from Elixir / BEAM, with full vendor
accelerator access on phones — Apple Neural Engine on iOS, MediaTek /
Qualcomm GPU+NPU HALs on Android. Same .tflite model file works on
all platforms.
{:nx_tflite_mob, "~> 0.0.3"}
Important: this is not an Nx backend
NxTfliteMob does not replace Nx.BinaryBackend / EMLX.Backend
/ NxVulkan.Backend. You can't do
Nx.global_default_backend(NxTfliteMob.Backend) — no such module
exists.
		Nx backend (EMLX, NxVulkan, NxEigen, BinaryBackend)	NxTfliteMob
	What you write	Nx.dot, Nx.conv, etc. (composable ops)	load_module(model_bytes) then call(handle, inputs)
	What runs	Each op dispatches to backend	The whole pre-compiled model graph executes through a vendor delegate
	Best for	Custom tensor math, arbitrary inference	Pre-trained models exported to .tflite — YOLO, MobileNet, MoveNet, etc.
	Apple Neural Engine	Indirect via MLX	Direct via Core ML delegate ⭐
	Android vendor NPU/GPU	Not available	Direct via NNAPI delegate ⭐
	Compose with Nx code	n/a — it IS Nx	Yes — input prep + output decode can use Nx.from_binary/3

Use NxTfliteMob when you have a pre-trained model. Use Nx backends
when you're writing arbitrary tensor code in Elixir.
30-second quickstart
# 1. Load a .tflite model.
tflite = File.read!("priv/yolov8n_float16.tflite")
{:ok, handle} = NxTfliteMob.load_module(tflite,
  delegate: "coreml",          # or "nnapi", "xnnpack"
  coreml_ane_only: false
)

# 2. Inference — bytes in, bytes out (model-specific shape + dtype).
{:ok, [output_bytes]} = NxTfliteMob.call(handle, [input_bytes])

# 3. Free the model when done.
:ok = NxTfliteMob.release_module(handle)
The model file is the standard TFLite FlatBuffer format. Anything
exportable from Ultralytics, MediaPipe, TF, JAX (via the AI Edge
Toolkit), or PyTorch (via ai-edge-torch) works.
See the YOLO walkthrough for a
complete worked example with input prep, NMS, and on-device perf
breakdown.
Two ways to use it
A. With Mob (mobile apps) — recommended
If you're building a Mob app, install via mob_dev's Igniter task:
mix mob.enable tflite

This runs once and:
	Adds {:nx_tflite_mob, "~> 0.0.3"} + {:nx, "~> 0.10"} to deps.
	Generates lib/<your_app>/tflite_init.ex with per-platform default
delegate opts (coreml on iOS, nnapi + mtk-gpu_shim on Android).
	Registers :tflite_nif in mob_dev's static-NIF table behind the
MOB_STATIC_TFLITE_NIF guard.
	Next mix mob.deploy --native automatically:	Downloads tensorflow-lite-2.16.1.aar (Android) or
TensorFlowLiteC-2.17.0.tar.gz (iOS) into ~/.mob/cache/
	Cross-compiles libtflite_nif.a per arch
	Links it into your app's main native binary
	Drops libtensorflowlite_jni.so into jniLibs/<abi>/ (Android)
or links the framework statically into the app binary (iOS)



Then in code:
{:ok, h} = NxTfliteMob.load_module(model_bytes, MyApp.TfliteInit.default_opts())
Requires mob_dev >= 0.5.9 from
hex.
B. Standalone (any Elixir app)
If you're not using Mob, build the NIF for your target host yourself:
git clone https://github.com/GenericJam/nx_tflite_mob.git
cd nx_tflite_mob

# Pick your target — see Makefile for options.
make android       # → priv/android_arm64/libtflite_nif.{so,a}
make ios_device    # → priv/ios_device/libtflite_nif.a
make ios_sim       # → priv/ios_sim/libtflite_nif.a
make mac           # → priv/mac/libtflite_nif.so (Mac-host tests)

The Mac build requires you to first build
libtensorflowlite_c.dylib from TF source — see
docs/build_mac_tflite.md. TFLite has no
Mac arm64 prebuilt distribution (Android + iOS are prebuilt, the
Makefile points at known cache locations for those).
Per-platform perf — measured on real hardware
	Device	Hardware	Model	Delegate	Latency
	iPhone SE 3rd gen	A15 + ANE	YOLOv8n FP16	Core ML → ANE	24 ms
	iPhone SE 3rd gen	A15	YOLOv8n INT8	XNNPACK CPU+NEON	37 ms
	Moto G Power 5G	Dimensity 7020 + PowerVR BXM-8-256	YOLOv8n INT8	NNAPI / mtk-gpu_shim	75-117 ms
	Moto G Power 5G	Dimensity 7020 + PowerVR BXM-8-256	YOLOv8n INT8	XNNPACK CPU+NEON	77 ms
	Moto G Power 5G	MediaTek APU/MDLA	YOLOv8n INT8	NNAPI / mtk-neuron_shim	355 ms (post-processing CPU fallback)

Numbers above are inference-call latency (median of 5 runs, after
warmup). Live-camera screens with input prep + output decode in BEAM
add 30-80 ms of overhead per frame — see the walkthrough for the
per-stage timing breakdown that took our Android live-YOLO loop from
0.5 FPS to 3.9 FPS.
The headline:
Apple Neural Engine via TFLite Core ML beats EMLX (~30 ms via MLX→ANE)
by ~20% on this model, because Core ML's compiler is more aggressive
about ANE op coverage than the MLX→Metal→ANE path. Same .tflite model
file, both numbers from the same iPhone.
What's in the package
		
	lib/nx_tflite_mob.ex	Elixir API — three public functions
	c_src/tflite_nif.c	C NIF wrapping the TFLite C API
	Makefile	Cross-compile per platform
	test/fixtures/add.bin	544-byte TFLite model (output = 3*input) for tests
	guides/	YOLO walkthrough, delegate selection guide
	docs/	Build recipes (Mac host build)

The NIF (libtflite_nif.{so,a}) is not in the published Hex
release — the package ships source, and the consumer's make (or
mob_dev's auto-build) produces per-platform binaries against
platform-appropriate TFLite distributions.
Architecture detail: how it works on each platform
Android
The Maven Central tensorflow-lite-2.16.1.aar ships
libtensorflowlite_jni.so for arm64-v8a + armv7a. mob_dev's
MobDev.TfliteDownloader extracts it into ~/.mob/cache/ and
MobDev.TfliteNif cross-compiles tflite_nif.c against the AAR's
headers via the Android NDK. The resulting libtflite_nif.a is
statically linked into your app's main native lib alongside the BEAM,
so the NIF init function is resolvable at app launch (no dlopen,
which Bionic's RTLD_LOCAL would block).
NNAPI is part of Android. The delegate routes through whichever
vendor HAL is installed — MediaTek's libmtk-gpu-shim.so, Qualcomm's
libqti-gpu.so, etc. accelerator_name selects which one.
iOS
CocoaPods CDN ships TensorFlowLiteC-2.17.0.tar.gz from dl.google.com
with .xcframework slices for ios-arm64 + ios-arm64_x86_64-simulator.
The framework binaries are unusual — they're MH_OBJECT (relocatable
object files, filetype=1), not MH_DYLIB. The linker pulls them
statically into your app's main Mach-O at build time. They do NOT need
to be embedded as runtime .framework bundles in the .app — trying
to embed them trips iOS install on missing Info.plist (CocoaPods
generates them) and then on "code signature version no longer
supported" (codesign only makes v3 sigs for MH_EXECUTE/MH_DYLIB).
Core ML delegate routes through Apple's Core ML framework, which
internally schedules to the Apple Neural Engine when ops are
supported.
Mac (host tests only)
TFLite has no Mac arm64 prebuilt distribution — we tried every channel
(pip install tflite-runtime, ai-edge-litert's wheel, MediaPipe's
wheel, TensorFlow's wheel, the iOS xcframework simulator slice) and
none yields a usable libtensorflowlite_c.dylib. Workaround: build it
from TF source via CMake (focused target, ~10-15 min one-time, cached
afterwards). See docs/build_mac_tflite.md.
Mac is host-tests-only; the dylib is not packaged into the published
Hex release. Production phone builds use the prebuilt AAR + xcframework.
Status
	Surface	State
	Hex release	✅ Published at hex.pm/packages/nx_tflite_mob
	HexDocs	✅ hexdocs.pm/nx_tflite_mob
	Android arm64	✅ via make android or mob_dev's mix mob.enable tflite
	iOS arm64 (device)	✅ via make ios_device or mob_dev
	iOS arm64 (simulator)	✅ via make ios_sim or mob_dev
	Mac arm64 (host tests)	✅ via make mac (build dylib first per docs/build_mac_tflite.md)
	Tests (16 integration + smoke)	✅ mix test against real .tflite
	End-to-end in Mob's running BEAM	✅ verified live: 24ms iPhone SE / 117ms Moto BXM

Versions pinned
	Distribution	Version	Source
	Android AAR	2.16.1	Maven Central org.tensorflow:tensorflow-lite
	iOS xcframework	2.17.0	dl.google.com (CocoaPods upstream)
	Mac (CMake-built)	2.16.1	TensorFlow source v2.16.1

These ship under different version pins because of upstream packaging
differences (Android's last AAR was 2.16.1; iOS's CocoaPod is at
2.17.0). The TFLite C API is binary-stable across this range — same
.tflite model file loads + runs identically on either version.
License
Apache 2.0. See LICENSE.
Acknowledgements
Built on top of:
	TensorFlow Lite — Apache 2.0, Google
tensorflow/lite
	Mob — the BEAM-on-device mobile framework this package was
built for
GenericJam/mob
	Nx — interop is optional but the type system makes pre/post
processing pleasant
elixir-nx/nx



  

    Changelog

All notable changes to nx_tflite_mob are documented here.
Format: Keep a Changelog.
Versioning: SemVer.

[0.0.4]
Changed
	Docs rewrite. This release is docs-only — no source-level
changes. Anyone landing on
hexdocs.pm/nx_tflite_mob
can now follow end-to-end without having to ask "is this an Nx
backend?" or "how do I decode the bytes?".	@moduledoc rewritten with a clear "this is NOT an Nx backend"
lede + per-platform delegate sections (Core ML, NNAPI, XNNPACK,
Metal-planned) + explicit input/output byte-layout tables for
common models + an "optional Nx interop" section showing how to
compose with EMLX.Backend / Nx.from_binary/3 if desired.
	README.md rewritten as the landing doc. Drops the stale "NIF
loads inside Mob's running BEAM ⏸ blocked" status (resolved
end-to-end in 0.0.3 via mob_dev's static-NIF integration). Adds
"Two ways to use it" — mix mob.enable tflite for Mob apps vs
make {android,ios_device,ios_sim,mac} for standalone Elixir
apps. Per-platform perf table now includes the measured iPhone
SE A15 / Moto BXM-8-256 numbers.
	New guide: guides/yolo_walkthrough.md — complete YOLOv8n
end-to-end walkthrough from model acquisition to bounding boxes,
with the full per-stage timing breakdown that took our Android
live-YOLO loop from 0.5 FPS to 3.9 FPS. Includes the pure-BEAM
INT8 decoder (130× faster than the equivalent Nx.BinaryBackend
decode) and the camera-format choice rationale per platform.
	New guide: guides/delegates.md — picking a delegate per
platform. Documents the "INT8 + Core ML doesn't work" trap
(0/256 nodes delegate), how to discover NNAPI accelerators on
Android, why mtk-neuron_shim NPU loses to mtk-gpu_shim GPU
for YOLO-class models (post-processing CPU fallback dominates),
and when to pick XNNPACK CPU even when GPU/NPU is available.
	docs() config now groups extras into "Guides" and "Build
recipes" sidebar sections.
	Tightened function @docs on load_module/2, call/2,
release_module/1 — explicit about input/output byte semantics +
error conditions.



Notes
	No code changes. The 0.0.3 NIF binary is bit-identical to the
0.0.4 NIF. Upgrade is a no-op for mix.lock.

[0.0.3]
Added
	Mac arm64 host build for testing. make mac produces
priv/mac/libtflite_nif.so linked against a locally-built
libtensorflowlite_c.dylib (since TFLite has no Mac arm64 prebuilt
distribution). Configurable via MAC_TFLITE_DIR (defaults to
~/.mob/cache/tflite-2.16.1-mac_arm64).
	Test suite: test/test_helper.exs + test/nx_tflite_mob_test.exs
with 16 tests covering module shape + package metadata (smoke tier,
always runs) and load_module / call / release_module / opt
normalisation (integration tier, auto-skipped when the host NIF
isn't built — keeps mix test green for users who only deploy to
phones).
	Test fixture: test/fixtures/add.bin — a 544-byte TFLite model
(output = 3*input, 1×8×8×3 float32) lifted from the upstream
tensorflow/lite/testdata/ set. Used by the integration tests to
prove the NIF executes a real model end-to-end on the host.
	docs/build_mac_tflite.md — reproducible recipe for building the
Mac libtensorflowlite_c.dylib from TF v2.16.1 source via CMake.
Documents the std::abs<T> libc++ patch and the
CMAKE_POLICY_VERSION_MINIMUM=3.5 env-var workaround for CMake-4
compatibility with TF's older cmake_minimum_required declarations.

Changed
	c_src/tflite_nif.c — __APPLE__ branches now refined with
TARGET_OS_IPHONE || TARGET_OS_SIMULATOR so Mac host builds skip the
iOS framework-style headers + Core ML delegate. Mac builds use the
same flat-path include layout as Android. No effect on iOS or
Android binaries.

Notes
	The Mac build is for host-side testing only. The dylib is not
packaged into the Hex release; production phone builds use the
prebuilt Android AAR + iOS xcframework as before.
	Tests run automatically in CI on macOS once the dylib is in
~/.mob/cache/. The cache step is a per-CI-runner one-time setup;
make mac reuses the cache afterwards.

[0.0.2]
Added
	iOS support. c_src/tflite_nif.c now compiles for ios_device
(arm64) and ios_sim (arm64) via xcrun, alongside the existing
Android arm64 path. The C NIF picks the right delegate per platform:	__ANDROID__ → NNAPI (mtk-gpu_shim, mtk-neuron_shim, etc.)
	__APPLE__ → Core ML (with optional coreml_ane_only for devices
with an Apple Neural Engine)
	xnnpack (default) on both


	Makefile targets: make ios_device, make ios_sim, make android,
make all_mobile. Each produces a per-arch libtflite_nif.{a,so}
under priv/<target>/.
	Framework-style includes on iOS (<TensorFlowLiteC/c_api.h> resolved
via -F-flagged search paths) vs. flat-path includes on Android
("tensorflow/lite/c/c_api.h" resolved via -I).

Measured on real hardware
	Device	Path	Inference
	Moto G Power 5G (BXM-8-256)	NNAPI / mtk-gpu_shim INT8	75-117 ms
	iPhone SE 3rd gen (A15)	Core ML → ANE (FP16 model)	24 ms

Notes
	iOS framework binaries (TensorFlowLiteC.framework/TensorFlowLiteC etc.)
ship as Mach-O MH_OBJECT (filetype=1), not MH_DYLIB. The linker
pulls them statically into the app's main binary at build time. Do
NOT embed them as runtime .framework bundles in the .app — it
trips iOS install twice (missing per-framework Info.plist, then
"code signature version no longer supported" since codesign only
produces v3 signatures for MH_EXECUTE/MH_DYLIB).
	Integration with the Mob
framework happens via mix mob.enable tflite in
mob_dev ≥ 0.5.8.

[0.0.1] — 2026-05-16
Initial release.
Added
	C NIF wrapping the TensorFlow Lite C API: load_module/2, call/2,
release_module/1.
	Android NNAPI delegate support with accelerator selection
(mtk-gpu_shim, mtk-neuron_shim, etc.).
	XNNPACK CPU path (default).
	Standalone Android bench CLI (scripts/bench_android/bench.c) that
hit 155 ms YOLOv8n via NNAPI mtk-gpu_shim on the Moto G Power
5G (2024) — the headline that prompted this package's existence.



  

    YOLOv8n end-to-end: from camera frame to bounding boxes

A complete worked example of using nx_tflite_mob to run live YOLOv8n
object detection on a Mob app, with the full timing breakdown of
where the milliseconds go.
The numbers in this guide are measured on:
	iPhone SE 3rd gen (Apple A15 + Neural Engine)
	Moto G Power 5G (2024) (MediaTek Dimensity 7020 + IMG PowerVR
BXM-8-256)

Why YOLO as the example
YOLO is hard. Real-time object detection chews through GPU/NPU
budget, the post-processing has tricky byte layouts, and the
.tflite model exists in several variants (INT8 / FP16 / FP32) with
different perf+accuracy tradeoffs. If you can do YOLO end-to-end, you
can do most things.
1. Get the model
We're using the Ultralytics YOLOv8n export — 3.4 MB INT8 (or 6 MB
FP16) trained on COCO.
pip install ultralytics
yolo export model=yolov8n.pt format=tflite int8=True imgsz=640
# → yolov8n_saved_model/yolov8n_full_integer_quant.tflite
# → yolov8n_saved_model/yolov8n_float16.tflite
# → yolov8n_saved_model/yolov8n_float32.tflite

The script also produces yolov8n_int8.tflite (different quant
flavour — symmetric int8 weights, asymmetric int8 activations) and
yolov8n_integer_quant.tflite (post-training quant without
calibration). For our purposes the full_integer_quant (uniform INT8
with calibration) is the right choice — best size + best NNAPI
delegation rate on Android.
Drop the model into your Mob app:
mkdir -p priv/yolo
cp yolov8n_saved_model/yolov8n_full_integer_quant.tflite priv/yolo/
cp yolov8n_saved_model/yolov8n_float16.tflite priv/yolo/
The COCO class labels you'll need for display:
@coco ~w(
  person bicycle car motorcycle airplane bus train truck boat
  traffic-light fire-hydrant stop-sign parking-meter bench bird cat
  dog horse sheep cow elephant bear zebra giraffe backpack umbrella
  handbag tie suitcase frisbee skis snowboard sports-ball kite
  baseball-bat baseball-glove skateboard surfboard tennis-racket
  bottle wine-glass cup fork knife spoon bowl banana apple sandwich
  orange broccoli carrot hot-dog pizza donut cake chair couch
  potted-plant bed dining-table toilet tv laptop mouse remote
  keyboard cell-phone microwave oven toaster sink refrigerator book
  clock vase scissors teddy-bear hair-drier toothbrush
)
2. Pick a model variant per platform
The platform-aware default-opts helper:
defmodule MyApp.YoloOpts do
  @doc "Returns `{model_file, opts}` for the best TFLite path on this device."
  def best do
    case :mob_nif.platform() do
      :ios ->
        # Core ML delegate hits the A15 Neural Engine. FP16 model gets
        # 56% of nodes delegated (24-25 ms median). INT8 + Core ML
        # delegates 0 nodes — don't use it on iOS.
        {"yolov8n_float16.tflite", [delegate: "coreml", coreml_ane_only: false]}

      :android ->
        # NNAPI's mtk-gpu_shim is the MediaTek GPU HAL on Dimensity
        # chips (~75-117 ms). Without an explicit accelerator name,
        # NNAPI defaults to mtk-neuron_shim NPU which is 5× SLOWER
        # for YOLO (post-processing falls back to CPU with cross-device
        # transfers). Always pass an explicit accelerator string.
        {"yolov8n_full_integer_quant.tflite",
         [delegate: "nnapi", accelerator: "mtk-gpu_shim", allow_fp16: true]}

      _ ->
        # Mac dev / Linux dev — XNNPACK CPU+SIMD path. Works
        # everywhere; surprisingly competitive (~75 ms on the same
        # Android chip).
        {"yolov8n_full_integer_quant.tflite", [delegate: "xnnpack"]}
    end
  end
end
The Android mtk-gpu_shim string is MediaTek-specific. For other OEMs
substitute:
	OEM	Accelerator string
	MediaTek (Dimensity)	mtk-gpu_shim, mtk-neuron_shim
	Qualcomm (Snapdragon)	qti-gpu, qti-dsp
	Samsung (Exynos)	samsung-gpu
	Google (Pixel)	google-edgetpu

See the delegates guide for how to discover
accelerators on a connected device.
3. Load the model
Load once at app start (or screen mount). Don't reload per-inference
— the delegate-init cost is ~few hundred ms on first load.
defmodule MyApp.YoloScreen do
  use Mob.Screen
  require Logger

  def mount(_params, _session, socket) do
    {model_file, opts} = MyApp.YoloOpts.best()
    priv = :code.priv_dir(:my_app) |> to_string()
    model_path = Path.join([priv, "yolo", model_file])

    handle =
      case NxTfliteMob.load_module(File.read!(model_path), opts) do
        {:ok, h} -> h
        {:error, reason} ->
          Logger.error("YOLO model load failed: \#{inspect(reason)}")
          nil
      end

    socket =
      socket
      |> Mob.Socket.assign(:tflite, handle)
      |> Mob.Socket.assign(:detections, [])
      |> Mob.Permissions.request(:camera)

    {:ok, socket}
  end
end
4. Hook up the camera
YOLOv8n's INT8 export expects 1×640×640×3 INT8 NHWC input. The FP16
export expects 1×640×640×3 FP32 NHWC input (FP16 is the weight
precision, not the input — the model casts internally).
The pre-frame conversion is the dominant non-inference cost — pick a
camera format that minimises it:
def handle_info({:permission, :camera, :granted}, socket) do
  format =
    case socket.assigns.quant do
      :int8 -> :bgra_u8   # 4 bytes/px; reorder + subtract 128 → INT8
      :fp16 -> :rgb_f32   # 12 bytes/px; pass straight through
    end

  socket =
    socket
    |> Mob.Camera.start_preview(facing: :back)
    |> Mob.Camera.start_frame_stream(
         width: 640, height: 640,
         format: format,
         facing: :back,
         throttle_ms: 40)

  {:noreply, socket}
end
:rgb_f32 is heavier (4.9 MB per frame) but converts to FP16 model
input with zero work — the model accepts FP32 directly. The 0 ms
conversion makes it the winning choice on iOS.
:bgra_u8 is lighter (1.6 MB per frame) and converts to INT8 via
pure byte arithmetic (drop alpha, swap BGR→RGB, subtract 128). About
15 ms in BEAM on the Moto BXM.
5. The inference call
def handle_info({:camera, :frame, %{bytes: bytes}}, socket) do
  t0 = System.monotonic_time(:millisecond)

  input = prepare_input(socket.assigns.quant, bytes)
  t_call = System.monotonic_time(:millisecond)

  case NxTfliteMob.call(socket.assigns.tflite, [input]) do
    {:ok, [output_bytes]} ->
      t_decode = System.monotonic_time(:millisecond)
      detections = decode_output(socket.assigns.quant, output_bytes)
      t_end = System.monotonic_time(:millisecond)

      Logger.debug(
        "yolo: conv=\#{t_call - t0}ms call=\#{t_decode - t_call}ms " <>
        "dec=\#{t_end - t_decode}ms detections=\#{length(detections)}"
      )

      {:noreply, Mob.Socket.assign(socket, :detections, detections)}

    {:error, reason} ->
      Logger.error("yolo: call failed: \#{inspect(reason)}")
      {:noreply, socket}
  end
end

# :int8 → camera bgra_u8 → model int8 NHWC (drop alpha, BGR→RGB, -128)
defp prepare_input(:int8, bgra_bytes) do
  for <<b, g, r, _a <- bgra_bytes>>, into: <<>> do
    <<r - 128::signed-8, g - 128::signed-8, b - 128::signed-8>>
  end
end

# :fp16 → camera rgb_f32 passes through (FP16 model takes FP32 input)
defp prepare_input(:fp16, rgb_f32_bytes), do: rgb_f32_bytes
6. Decode the output
YOLOv8n's output is 1×84×8400 (4 box coords + 80 class probs ×
8400 anchors). On INT8 models it's INT8 bytes with quantization
scale + zero_point. On FP16/FP32 models it's FP32 with box coords
normalised to [0, 1] (Ultralytics convention).
Skip Nx for this — the BinaryBackend argmax over {80, 8400} was
1700 ms in our profiling; a pure-BEAM byte loop is 13 ms:
@n_anchors 8400
@n_classes 80
@scale 0.00659          # INT8 quant scale (from the .tflite metadata)
@zero -128              # INT8 zero_point

defp decode_output(:int8, bytes) do
  conf_int8_threshold = trunc(0.25 / @scale) + @zero

  Enum.reduce(0..(@n_anchors - 1), [], fn anchor, acc ->
    {class_id, max_v} = max_class_int8(bytes, anchor, 0, 0, -129)

    if max_v > conf_int8_threshold do
      [build_candidate_int8(bytes, anchor, class_id, max_v) | acc]
    else
      acc
    end
  end)
  |> Enum.sort_by(& &1.confidence, :desc)
  |> Enum.take(20)
  |> nms(0.45)
end

# Scan 80 class bytes for `anchor`. Class probs occupy
# bytes[4*8400 + cls*8400 + anchor] for cls in 0..79.
defp max_class_int8(_bytes, _anchor, @n_classes, max_cls, max_v),
  do: {max_cls, max_v}

defp max_class_int8(bytes, anchor, cls, max_cls, max_v) do
  raw = :binary.at(bytes, 4 * @n_anchors + cls * @n_anchors + anchor)
  v = if raw > 127, do: raw - 256, else: raw  # uint8 -> int8

  if v > max_v do
    max_class_int8(bytes, anchor, cls + 1, cls, v)
  else
    max_class_int8(bytes, anchor, cls + 1, max_cls, max_v)
  end
end

defp build_candidate_int8(bytes, anchor, class_id, max_v) do
  cx = int8_to_f32(:binary.at(bytes, anchor))
  cy = int8_to_f32(:binary.at(bytes, @n_anchors + anchor))
  w  = int8_to_f32(:binary.at(bytes, 2 * @n_anchors + anchor))
  h  = int8_to_f32(:binary.at(bytes, 3 * @n_anchors + anchor))

  %{
    class_id: class_id,
    confidence: int8_to_f32(max_v),
    x1: cx - w / 2, y1: cy - h / 2,
    x2: cx + w / 2, y2: cy + h / 2
  }
end

defp int8_to_f32(b) do
  signed = if b > 127, do: b - 256, else: b
  (signed - @zero) * @scale
end

# Non-max suppression. Already top-20, so the O(N²) loop is cheap.
defp nms([], _), do: []
defp nms([best | rest], iou_th) do
  rest = Enum.reject(rest, &(iou(best, &1) > iou_th))
  [best | nms(rest, iou_th)]
end

defp iou(a, b) do
  ix1 = max(a.x1, b.x1); iy1 = max(a.y1, b.y1)
  ix2 = min(a.x2, b.x2); iy2 = min(a.y2, b.y2)
  inter = max(0, ix2 - ix1) * max(0, iy2 - iy1)
  union = (a.x2 - a.x1) * (a.y2 - a.y1) + (b.x2 - b.x1) * (b.y2 - b.y1) - inter
  if union <= 0, do: 0.0, else: inter / union
end
The FP32 decoder is the same shape but reads 4-byte floats and
skips dequantization. Box coords need a × 640 scale because
Ultralytics FP32 normalizes them to [0, 1].
7. Where the milliseconds go
End-to-end per-frame breakdown on a Moto G Power 5G (best case
after warmup):
	Stage	Time	What it does
	prepare_input/2 (bgra→int8)	15 ms	BEAM bitstring comprehension over 410k pixels
	NxTfliteMob.call/2	75-117 ms	TFLite + NNAPI → MediaTek GPU HAL
	decode_output/2 + NMS	13 ms	Pure-BEAM byte scan over 8400×84
	Total	~110-150 ms	6-9 FPS sustained

On iPhone SE 3rd gen:
	Stage	Time	What it does
	prepare_input/2 (rgb_f32 passthrough)	0 ms	Pass camera bytes straight to the FP16 model
	NxTfliteMob.call/2	25-40 ms	TFLite + Core ML → Apple Neural Engine
	decode_output/2 + NMS	60-85 ms	Pure-BEAM FP32 decode (slower than INT8 — 4-byte float reads vs 1-byte)
	Total	~90-125 ms	8-11 FPS sustained

Notable: the FP32 decoder is the iOS bottleneck. The Apple Neural
Engine call itself is 25 ms — the same as the standalone bench. The
remaining 60-85 ms is BEAM byte processing of the f32 output tensor.
Future optimization: a tiny NIF helper for the float-scan
(:binary.at + 4-byte pattern matches are the slow part).
8. Optional Nx interop
If you prefer Nx for the prep/decode (slower but more declarative):
# Input prep via Nx (FP16 path on iOS — no quant math needed)
input =
  camera_rgb_f32_bytes
  |> Nx.from_binary(:f32, backend: Nx.BinaryBackend)
  |> Nx.reshape({1, 640, 640, 3})
  |> Nx.to_binary()

{:ok, [out]} = NxTfliteMob.call(handle, [input])

# Output decode via Nx
out
|> Nx.from_binary(:f32, backend: Nx.BinaryBackend)
|> Nx.reshape({1, 84, 8400})
|> ...  # box decode + NMS
The backend: option is critical — Nx.BinaryBackend is pure Elixir
and very slow for million-element ops. Using EMLX.Backend for the
Nx side on iOS speeds this up substantially:
input =
  camera_rgb_f32_bytes
  |> Nx.from_binary(:f32, backend: {EMLX.Backend, device: :gpu})
  |> ...
This composes — EMLX.Backend for the Nx-side math, NxTfliteMob
for the TFLite-side model call. Two distinct compute paths in the
same screen.
See also
	delegates guide — picking the right delegate per
platform, accelerator discovery
	docs/build_mac_tflite.md — building
libtensorflowlite_c.dylib from source for Mac host tests
	The
LiveYoloScreen
in the nxeigen_probe reference app for a working version of all
of the above



  

    Picking a delegate

A TFLite "delegate" is the runtime that actually executes the model
graph. TFLite ships several; picking the right one for your platform
	model combination is the single biggest perf lever.

This guide lays out the decision matrix and explains the quirks we
discovered measuring real hardware.
Quick decision tree
Are you on iOS (any iPhone/iPad)?
├── Model is FP16 or FP32?  → delegate: "coreml"        (24 ms iPhone SE A15)
└── Model is INT8?          → delegate: "xnnpack"       (27 ms iPhone SE A15)
                              (NOT coreml — 0/256 nodes delegate)

Are you on Android?
├── MediaTek Dimensity?     → delegate: "nnapi", accelerator: "mtk-gpu_shim"
├── Qualcomm Snapdragon?    → delegate: "nnapi", accelerator: "qti-gpu"
├── Pixel?                  → delegate: "nnapi", accelerator: "google-edgetpu"
├── Unknown OEM?            → delegate: "xnnpack"       (always works)
└── Want to discover?       → see "Discovering NNAPI accelerators" below

Mac / Linux dev host?       → delegate: "xnnpack"
Per-delegate detail
xnnpack — CPU+SIMD
Bundled into TFLite. Cross-platform. Default when no other delegate
is set explicitly.
Highly-optimised: tuned ARM NEON / Intel AVX kernels, INT8 + FP32 +
quantized-int8 paths.
	Pro	Con
	Works everywhere TFLite runs	No GPU / NPU
	Reproducible numbers (no thermal throttle)	Slower than accelerator paths when those work
	No vendor-driver dependencies	

Options:
	num_threads: — CPU thread count (default 6). Up to physical core
count helps; oversubscription hurts.

On the Moto G Power 5G (Dimensity 7020), XNNPACK matched the GPU
delegate at 77 ms for YOLOv8n. On modern phones with strong CPU
cores, XNNPACK is a competitive default.
nnapi — Android Neural Networks API
Android's neural-net dispatch layer. Each device's vendor ships a
HAL driver (libmtk-gpu-shim.so, libqti-gpu.so, etc.); NNAPI picks
one based on the accelerator name (or by default, badly — see
below).
Options:
	accelerator: — vendor HAL name string. Always pass this
explicitly. NNAPI's auto-selection on at least one MediaTek device
picks the NPU which is 5× SLOWER than the GPU for YOLO-class models.
	allow_fp16: — let the HAL promote FP32 ops to FP16 (default
true). Lossy but typically fine for inference.

Discovering NNAPI accelerators on a connected device
The standalone bench CLI (in scripts/bench_android/) has a
list-nnapi mode:
adb push scripts/bench_android/bench /data/local/tmp/
adb push ~/.mob/cache/tflite-2.16.1-android_arm64/jni/arm64-v8a/libtensorflowlite_jni.so /data/local/tmp/
adb shell 'cd /data/local/tmp && LD_LIBRARY_PATH=. ./bench list-nnapi'

Output is a list of accelerator names available on this device:
mtk-gpu_shim
mtk-neuron_shim
nnapi-reference  (CPU emulation — slow)
Known accelerator perf rankings (YOLOv8n on Moto G Power 5G)
	Accelerator	Median
	mtk-gpu_shim	75-117 ms (best — MediaTek's PowerVR HAL)
	xnnpack CPU	77-91 ms (tied with GPU; deterministic)
	mtk-neuron_shim	355 ms (NPU — slower because YOLO post-processing falls back to CPU)
	nnapi-reference	358 ms (CPU emulation — never use)
	nnapi (no accelerator)	358 ms (defaults to mtk-neuron_shim — never do this)

The NPU loses despite being a "real" neural-net accelerator because
YOLO has concat + reshape ops in its post-processing that aren't
in the NPU's supported set. TFLite falls back to CPU for those ops
mid-graph, with cross-device buffer transfers between each fallback.
The transfer overhead swamps any per-op NPU speedup.
A model designed end-to-end for the APU (no reshape/concat in the
inference graph) would land much faster on mtk-neuron_shim. YOLOv8n
as exported doesn't fit.
coreml — Apple Core ML
Routes the delegated portion of the graph through Apple's Core ML
framework, which internally schedules to the Apple Neural Engine
when ops are supported on devices that have one (A11+ iPhones).
Options:
	coreml_ane_only: — when true, load_module/2 returns
{:error, _} instead of falling back to CPU on devices without an
ANE. Useful for "ANE-only or skip" logic. Default false.

Op-coverage caveat — the INT8 trap
Don't use Core ML with INT8 models. The Ultralytics
yolov8n_full_integer_quant.tflite export uses INT8 quantization ops
that Core ML's tooling doesn't translate to ANE primitives. The
result: 0 out of 256 nodes delegated, and the whole model falls
back to CPU which is slower than just running XNNPACK directly.
For Core ML you want the FP16 or FP32 model variant:
	Model	Core ML delegation rate	Latency (iPhone SE A15)
	INT8	0/256 (0%) — full CPU fallback	45 ms (don't use)
	FP16	214/385 (56%)	23-25 ms
	FP32	214/254 (84%)	24-25 ms

FP16 and FP32 hit the same wall-clock because the delegated portion
is the same (214 conv-shaped ops). FP16 wins on bundle size (~6 MB
vs ~12 MB).
The 30% of nodes that fall to CPU on FP16 are the post-processing
ops (concat / reshape / NMS-prep) — same shape as the Android NPU
problem. Core ML handles the boundary more gracefully than NNAPI
NPU does (cheap shared-memory transitions on Apple silicon), which
is why this works at all.
metal — Apple Metal GPU (planned)
TFLite ships TensorFlowLiteCMetal.xcframework with a Metal GPU
delegate, but the current NIF doesn't expose it as a delegate:
option. PR welcome.
Core ML is usually faster than Metal on Apple Silicon since it can
pick ANE for supported ops + Metal as a fallback. Metal-only is
mainly useful for older devices without an ANE.
Comparing the paths on the same device
Same iPhone SE 3rd gen A15, same .tflite model files, varying the
delegate:
	Variant	Delegate	Delegation	Min / Median / Max
	INT8	xnnpack	n/a (CPU+NEON)	27 / 36 / 37 ms
	INT8	coreml	0/256 (full fallback)	36 / 39 / 42 ms
	FP16	xnnpack	n/a (CPU+NEON)	86 / 98 / 265 ms
	FP16	coreml	214/385 (56%)	23 / 25 / 26 ms
	FP32	coreml	214/254 (84%)	24 / 24 / 25 ms

The standout: FP16 + Core ML wins at 25 ms median. Half the
bundle of FP32 with identical wall-clock. The CPU+NEON XNNPACK path
is impressive at 36 ms — for context, our standalone bench
measurements show it consistently within 30 ms of the GPU/ANE paths
on modern phones.
Composing with Nx backends
TFLite delegates handle the model graph. The pre/post-processing in
your Elixir code is separate compute that you can route to a
different backend:
# Input prep on EMLX (Metal GPU on iOS) — useful for batch
# transformations, scaling, normalization.
input_bytes =
  camera_bytes
  |> Nx.from_binary(:f32, backend: {EMLX.Backend, device: :gpu})
  |> Nx.reshape({1, 640, 640, 3})
  |> Nx.divide(255.0)
  |> Nx.to_binary()

# Model inference on TFLite + Core ML → ANE
{:ok, [out]} = NxTfliteMob.call(handle, [input_bytes])

# Output decode on EMLX again
out
|> Nx.from_binary(:f32, backend: {EMLX.Backend, device: :gpu})
|> Nx.reshape({1, 84, 8400})
|> ...
Two distinct compute paths, one screen. The TFLite delegate doesn't
care what backend your Nx code uses — it sees only the bytes you
hand to call/2.
When xnnpack is the right answer even when GPU/NPU is available
	Deterministic numbers (CPU paths don't thermal-throttle as
aggressively)
	Cold-start (delegate init for Core ML / NNAPI is 100-500 ms)
	Tiny models (the delegate dispatch overhead dominates inference
for sub-ms models)
	Cross-platform parity for tests



  

    Building libtensorflowlite_c.dylib for Mac arm64

TFLite has no Mac arm64 prebuilt distribution. To run host-side
integration tests (make mac && mix test), build the C library from
TensorFlow's source via CMake — this is a one-time ~10-15 min step;
the output dylib is then cached at ~/.mob/cache/.
Prereqs
brew install cmake      # CMake 3.16+ required; tested with 4.2.1
# Apple Clang (any recent Xcode CLT version)

Build
# 1. Shallow-clone TensorFlow at the version we pin everywhere else.
git clone --depth=1 --branch v2.16.1 \
  https://github.com/tensorflow/tensorflow.git ~/code/tensorflow_src

# 2. Patch one source file — Apple Clang 21+ libc++ dropped the
#    explicit-instantiation form of std::abs<T>. Replace
#    `std::abs<float>` with a lambda; same for std::abs<int32_t>.
#    The exact patch is documented inline below.

In tensorflow/lite/kernels/elementwise.cc, around the AbsEval
function (line ~290), change:
case kTfLiteFloat32:
  return EvalImpl<float>(context, node, std::abs<float>, type);
// ...
case kTfLiteInt32:
  return EvalImpl<int32_t>(context, node, std::abs<int32_t>, type);
to:
case kTfLiteFloat32:
  return EvalImpl<float>(
      context, node, [](float x) { return std::abs(x); }, type);
// ...
case kTfLiteInt32:
  return EvalImpl<int32_t>(
      context, node, [](int32_t x) { return std::abs(x); }, type);
# 3. Configure + build with the CMake-4 compat env var (propagates to
#    sub-projects' ExternalProject downloads, several of which use
#    `cmake_minimum_required(VERSION 2.x or 3.0)` and would otherwise
#    fail).
mkdir -p /tmp/tflite_build_mac && cd /tmp/tflite_build_mac
export CMAKE_POLICY_VERSION_MINIMUM=3.5
cmake ~/code/tensorflow_src/tensorflow/lite/c -DCMAKE_BUILD_TYPE=Release
cmake --build . --target tensorflowlite_c -j$(sysctl -n hw.ncpu)

Output: /tmp/tflite_build_mac/libtensorflowlite_c.dylib (~5 MB).
Install to cache
make mac looks for the dylib + headers under
~/.mob/cache/tflite-2.16.1-mac_arm64/ by default (override with
MAC_TFLITE_DIR=…). Install layout:
mkdir -p ~/.mob/cache/tflite-2.16.1-mac_arm64/{lib,include/tensorflow/lite/{c,core/{c,async/c}}}

cp /tmp/tflite_build_mac/libtensorflowlite_c.dylib \
   ~/.mob/cache/tflite-2.16.1-mac_arm64/lib/

cp ~/code/tensorflow_src/tensorflow/lite/*.h \
   ~/.mob/cache/tflite-2.16.1-mac_arm64/include/tensorflow/lite/

cp ~/code/tensorflow_src/tensorflow/lite/c/*.h \
   ~/.mob/cache/tflite-2.16.1-mac_arm64/include/tensorflow/lite/c/

cp ~/code/tensorflow_src/tensorflow/lite/core/*.h \
   ~/.mob/cache/tflite-2.16.1-mac_arm64/include/tensorflow/lite/core/

cp ~/code/tensorflow_src/tensorflow/lite/core/c/*.h \
   ~/.mob/cache/tflite-2.16.1-mac_arm64/include/tensorflow/lite/core/c/

cp ~/code/tensorflow_src/tensorflow/lite/core/async/c/*.h \
   ~/.mob/cache/tflite-2.16.1-mac_arm64/include/tensorflow/lite/core/async/c/

Verify
cd ~/code/nx_tflite_mob
make mac          # produces priv/mac/libtflite_nif.so (~50 KB)
cp priv/mac/libtflite_nif.so priv/native/libtflite_nif.so
mix test          # 16 tests, 0 failures

Why this is host-test-only
The dylib produced here links against @rpath/libtensorflowlite_c.dylib
with the rpath pointing into ~/.mob/cache/. It's not packaged into the
Hex release — production phone builds use the prebuilt Android AAR /
iOS xcframework. Mac-host builds exist solely so mix test can exercise
the NIF locally before shipping.
Future work
When LiteRT (Google's TFLite successor) lands on Maven Central +
CocoaPods, the per-platform make {android, ios_device, ios_sim}
targets become candidates for migration. As of v2.1.4 LiteRT only ships
via PyPI (Mac/Linux) and GitHub-Releases-headers + Bazel — no
Android/iOS-consumable artifacts, so the cross-platform story isn't
ready for migration. See the nx_tflite_mob repo's CHANGELOG.md
"future work" notes.

Call TensorFlow Lite models from Elixir / BEAM, with full vendor
accelerator access on phones — Apple Neural Engine on iOS, MediaTek
/ Qualcomm GPU+NPU HALs on Android.
This is NOT an Nx.Backend
NxTfliteMob does not replace Nx.BinaryBackend, EMLX.Backend,
NxVulkan.Backend, etc. There is no NxTfliteMob.Backend module
to set via Nx.global_default_backend/1.
TFLite executes pre-compiled model graphs (.tflite files)
end-to-end through vendor-optimised delegates. The whole graph
stays opaque so the delegate can fuse + schedule it for ANE / GPU
/ NPU. You don't compose your own ops here — you call a pre-trained
model.
Use NxTfliteMob when you have a pre-trained model to run. Use
Nx backends when you're writing arbitrary tensor math in Elixir.
Both can coexist in the same app.
API surface
Three functions: load_module/2, call/2, release_module/1.
iex> tflite = File.read!("priv/yolov8n_float16.tflite")
iex> {:ok, handle} = NxTfliteMob.load_module(tflite,
...>   delegate: "coreml", coreml_ane_only: false)
iex> {:ok, [output_bytes]} = NxTfliteMob.call(handle, [input_bytes])
iex> NxTfliteMob.release_module(handle)
:ok
See the YOLO walkthrough for a complete
end-to-end example with input prep, inference, and output decode.
Delegate options
The delegate opt selects how the model graph runs. Per-platform
recommendations (see also the delegates guide):
Android — delegate: "nnapi"
NNAPI is Android's neural-net dispatch API. It picks a vendor HAL
driver based on the accelerator name:
	accelerator: value	What it routes to
	"mtk-gpu_shim"	MediaTek's GPU HAL — fastest for YOLO on Dimensity chips
	"mtk-neuron_shim"	MediaTek's APU/NPU — only worthwhile if your graph is pure conv (no concat/reshape post-processing — TFLite falls back to CPU for those, transfer overhead dominates)
	"qti-gpu"	Qualcomm Snapdragon GPU
	"google-edgetpu"	Pixel TPU
	nil (no key)	NNAPI auto-picks — often the WRONG choice for YOLO (defaults to NPU on MediaTek, which is 5× slower)

Discover available accelerators on a connected device with
adb shell + the standalone bench CLI's list-nnapi mode (see
the package's scripts/bench_android/).
Other Android opts:
	num_threads: — XNNPACK CPU thread count (default 6)
	allow_fp16: — let NNAPI run FP32 ops in FP16 (default true)

iOS — delegate: "coreml"
Core ML routes the delegated portion through Apple's Core ML
framework, which internally schedules to the Apple Neural Engine
when ops are supported. For YOLOv8n FP16, ~56% of nodes delegate
to the ANE on an iPhone SE 3rd gen A15 (the rest fall to CPU via
XNNPACK), hitting 24 ms per inference.
Caveats:
	INT8 + Core ML doesn't work. Core ML's tooling doesn't
understand the Ultralytics INT8 quant flavour — 0/256 nodes
delegate. Use the FP16 model variant for Core ML.
	coreml_ane_only: (default false) — when true, the
delegate returns nil instead of falling back to CPU on
devices without an ANE. Useful for "ANE-only or skip" logic;
irrelevant on A11+ devices where the ANE is always present.

iOS — delegate: "metal" (planned)
TFLite ships TensorFlowLiteCMetal.xcframework for Metal GPU
inference but the current NIF doesn't expose it as a delegate:
option yet. PR welcome. Core ML is usually faster anyway on Apple
Silicon devices (Core ML can pick GPU when ANE ops are unsupported).
XNNPACK CPU — delegate: "xnnpack" (default)
Bundled into TFLite. Highly-optimised CPU+SIMD path. Default when
no other delegate is set. Surprisingly competitive on modern phones
— ~77 ms on the Moto G Power 5G (tied with the GPU path) and 27-37
ms on iPhone SE 3rd gen A15. Use this when:
	You're on a device without GPU/NPU acceleration
	The vendor delegate fails to delegate (e.g. INT8 + Core ML)
	You want deterministic, reproducible numbers (CPU paths don't
thermal-throttle as aggressively as GPUs)

Input + output byte layout
call/2 is raw-bytes-in, raw-bytes-out. The byte layout is
model-specific — you have to match what the .tflite model
expects.
Inspect a model's expected shape/dtype via mix Python helpers or
TFLite's flatc tool. Or :erlang.load_nif/2 an inspector NIF
built against TFLite's TfLiteInterpreterGetInputTensor —
exposing this in the Elixir API is on the roadmap.
Common shapes:
	Model	Input	Output
	YOLOv8n INT8 (Ultralytics full_integer_quant)	1×640×640×3 INT8 NHWC (1228800 bytes)	1×84×8400 INT8 (705600 bytes)
	YOLOv8n FP16 (Ultralytics float16)	1×640×640×3 FP32 NHWC (4915200 bytes — the FP16 model accepts FP32 input that's cast internally)	1×84×8400 FP32 normalised (2822400 bytes)
	YOLOv8n FP32	1×640×640×3 FP32 NHWC	1×84×8400 FP32
	MobileNetV2 (ImageNet)	1×224×224×3 FP32 NHWC	1×1001 FP32 (class logits)

See the YOLO walkthrough for the layout-aware decoder we use in
production (pure-BEAM, 13 ms for the full INT8 NMS pass).
Where Nx fits in (optionally)
You CAN use Nx tensors on either side of call/2. It's optional —
bytes-in/bytes-out is the canonical interface.
Input prep with Nx:
input_bytes =
  camera_frame_f32_binary
  |> Nx.from_binary(:f32)
  |> Nx.reshape({1, 640, 640, 3})
  |> Nx.as_type(:s8)      # quantize for INT8 model
  |> Nx.to_binary()

{:ok, [out]} = NxTfliteMob.call(handle, [input_bytes])
Output decode with Nx:
detections =
  out
  |> Nx.from_binary(:s8)
  |> Nx.reshape({1, 84, 8400})
  |> Nx.as_type(:f32)
  |> Nx.multiply(scale)
  |> Nx.subtract(zero_point)
  |> extract_detections()
In practice we bypass Nx for performance-critical decoding —
Nx.BinaryBackend for an argmax across {80, 8400} is 1700 ms;
a pure-BEAM :binary.at/2 loop is 13 ms (130× faster). See
NxeigenProbe.LiveYoloScreen for the pure-BEAM decoder pattern.
Using with Mob
If you're building a Mob app,
the easiest path is mob_dev's Igniter task:
mix mob.enable tflite
This adds the dep + generates a per-platform default-opts helper
and registers the NIF in mob_dev's static-NIF table. Requires
mob_dev >= 0.5.9. See mob_dev's
mob.enable docs
for details.
After mix mob.enable tflite, the auto-generated helper picks
delegate opts per platform:
{:ok, h} = NxTfliteMob.load_module(model_bytes,
             MyApp.TfliteInit.default_opts())
Building from source (non-Mob)
See the package's Makefile — targets android, ios_device,
ios_sim, mac. Each requires platform-appropriate TFLite
distribution (cached at ~/.mob/cache/ by mob_dev's downloader, or
per-target overrides for standalone builds).
Mac builds require building libtensorflowlite_c.dylib from TF
source first — TFLite has no Mac arm64 prebuilt. See
docs/build_mac_tflite.md in the repo.

        module_handle()

      


        Opaque handle to a loaded TFLite model. Pass to call/2 and free
with release_module/1. Closed handles also get freed when garbage
collected, but explicit release is recommended for short-lived
inferences.



    





  
    Functions
  


    
      
        call(handle, inputs)

      


        Run inference on a loaded model.



    


    
      
        load_module(model_bytes, opts \\ [])

      


        Load a TFLite model from raw .tflite FlatBuffer bytes.



    


    
      
        release_module(handle)

      


        Free the model + delegate + interpreter held by handle.

          @type module_handle() :: reference()


      


Opaque handle to a loaded TFLite model. Pass to call/2 and free
with release_module/1. Closed handles also get freed when garbage
collected, but explicit release is recommended for short-lived
inferences.

          @spec call(module_handle(), [binary()]) ::
  {:ok, [binary()]} | {:error, String.t() | charlist()}


      


Run inference on a loaded model.
inputs is a list of binaries — one per input tensor in the model's
declared input order. Each binary must match the model's expected
shape × dtype byte layout exactly (1×640×640×3 INT8 = 1228800 bytes
for YOLOv8n full_integer_quant, for example).
Returns {:ok, outputs} where outputs is a list of binaries — one
per output tensor, also in declared order. Decode each according to
the model's documented output layout.
Examples
# YOLOv8n INT8 — 1×640×640×3 INT8 input, 1×84×8400 INT8 output
input = <<…1228800 INT8 bytes…>>
{:ok, [output]} = NxTfliteMob.call(handle, [input])
true = byte_size(output) == 705600
Errors
Returns {:error, message} for:
	Input list length doesn't match the model's input-tensor count
	Any input binary's size doesn't match the model's expected size
	The model's TfLiteInterpreterInvoke returns non-OK status

          @spec load_module(
  binary(),
  keyword()
) :: {:ok, module_handle()} | {:error, String.t() | charlist()}


      


Load a TFLite model from raw .tflite FlatBuffer bytes.
Returns {:ok, handle} on success or {:error, message} if the
bytes aren't a valid TFLite model or delegate creation fails.
Options
All options are documented in detail in the moduledoc:
	:delegate (string) — "xnnpack" (default), "nnapi" (Android),
"coreml" (iOS)
	:accelerator (string) — vendor accelerator name for NNAPI
(e.g. "mtk-gpu_shim")
	:num_threads (integer) — XNNPACK CPU thread count (default 6)
	:allow_fp16 (boolean) — NNAPI FP32→FP16 promotion (default
true)
	:coreml_ane_only (boolean) — Core ML requires ANE (default
false — falls back to CPU/GPU)

Examples
# XNNPACK CPU (cross-platform default)
{:ok, h} = NxTfliteMob.load_module(tflite_bytes, [])

# Android NNAPI → MediaTek GPU HAL
{:ok, h} = NxTfliteMob.load_module(tflite_bytes,
             delegate: "nnapi",
             accelerator: "mtk-gpu_shim",
             allow_fp16: true)

# iOS Core ML → ANE
{:ok, h} = NxTfliteMob.load_module(tflite_bytes,
             delegate: "coreml",
             coreml_ane_only: false)

  



  
    
      
    
    
      release_module(handle)



        
          
        

    

  


  

      

          @spec release_module(module_handle()) :: :ok


      


Free the model + delegate + interpreter held by handle.
Idempotent — calling on an already-released handle returns :ok
(the underlying resource is zero'd and re-releasing is a no-op).
Resources are also freed on GC if release_module/1 isn't called,
but explicit release is recommended for tight loops or short-lived
inferences to keep memory predictable.

